```
;*************************************************************
;                                                            *
;       PROGRAM ID:     SYSTEM BOOTSTRAP DRIVER              *
;                                                            *
;*************************************************************
;                                                            *
;       PROPERTY OF:    JADE COMPUTER PRODUCTS               *
;                       4901 W. ROSECRANS BLVD.              *
;                       HAWTHORNE, CALIFORNIA                *
;                       90250,   U.S.A.                      *
;                                                            *
;*************************************************************
;                                                            *
;       VERSION:        2.2                                  *
;                                                            *
;*************************************************************
; THE  SYSTEM BOOTSTRAP DRIVER IS ONE OF TWO MODULES         *
; THAT MAKE UP THE  SYSTEM  RESIDENT BOOTSTRAP. THIS         *
; MODULE IS TO BE EXECUTED BY  THE SYSTEM PROCESSOR.         *
; DURING EXECUTION, THIS MODULE PERFORMS A BLOCK MOVE        *
; OF THE SECOND MODULE  (BOOT INJECTION MODULE) INTO         *
; THE DOUBLE D CONTROLLER MEMORY.  A SUCCESSFUL BOOT         *
; OPERATION BY THE DOUBLE D WILL LEAVE DCM IN BANK 0         *
; AND BIOS IN  BANK 1.  THE REMAINDER OF THIS MODULE         *
; THEN MOVES THE  BIOS  IMAGE TO  THE  PROPER SYSTEM         *
; ADDRESS AND  JUMPS TO THE  BIOS COLD START ENTRY.          *
;************************************************* SK ***
```

```
                        ;*************************************************************
                        ; DOUBLE D HARDWARE PARAMETERS.  PLEASE NOTE THIS      *
                        ; SECTION CONTAINS CONDITIONAL STATEMENTS.            *
                        ;*************************************************************

0043                    D.PORT  ==      043H    ;DOUBLE D PORT ADDRESS.
0001                    TRUE    ==      1       ;TRUE IS A ONE.
0000                    FALSE   ==      0       ;FALSE IS A ZERO.
0001                    REV.B   ==      TRUE    ;SET TRUE FOR REV B BOARDS.
0000                    REV.C   ==      FALSE   ;SET TRUE FOR REV C BOARDS.
0000                    MA10    ==      FALSE   ;TRUE IF MA10 JUMPED (REV-B).

                        .IFG    REV.B, [
0002                    DS.HLT  ==      002H    ;STATUS PORT HALT INDICATOR.
000C                    DS.ASW  ==      00CH    ;STATUS PORT ADDR SW MASK.
E400                    D.BASE  =       0E400H  ;SYSTEM WINDOW BASE ADDRESS]

                        .IFG    MA10, [
                        D.BASE  ==      0E000H  ;SYSTEM WINDOW BASE ADDRESS]

                        .IFG    REV.C, [
                        DS.HLT  ==      001H    ;STATUS PORT HALT INDICATOR.
                        DS.ASW  ==      00EH    ;STATUS PORT ADDR SW MASK.
                        D.BASE  ==      0E000H  ;SYSTEM WINDOW BASE ADDRESS]

                        ;*************************************************************
                        ; BOOTSTRAP INJECTION MODULE PARAMETERS (ALTERABLE)    *
                        ;*************************************************************

0200                    IM.ADR  ==      0200H   ;BOOT INJECTION MODULE ADDRESS.
00C0                    IM.SZE  ==      00C0H   ;BOOT INJECTION MODULE SIZE.

                        ;*************************************************************
                        ; BOOTSTRAP LINKAGE ADDRESS.                           *
                        ;*************************************************************

0080                    BSTACK  ==      0080H   ;BOOTSTRAP TOP OF STACK.
0040                    D.ADDR  ==      0040H   ;DOUBLE D ADDRESS POINTER.
0377                    BL.DCS  ==      0377H   ;DCM DISK CONTROLLER STATUS.
0378                    BL.ADR  ==      0378H   ;DCM LOAD AND JUMP ADDRESS.
037A                    BL.BSZ  ==      037AH   ;DCM BLOCK LOAD SIZE.

                        ;*************************************************************
                        ; DOUBLE D HARDWARE COMMANDS                           *
                        ;*************************************************************

0080                    DC.BGN  ==      080H    ;RESET Z80A AND EXECUTE.
0001                    DC.MRQ  ==      001H    ;REQUEST MEMORY WINDOW.
0000                    DC.MRT  ==      000H    ;RELEASE MEMORY WINDOW.
0001                    DC.MB0  ==      001H    ;SELECT MEMORY BANK 0.
0003                    DC.MB1  ==      003H    ;SELECT MEMORY BANK 1.
0002                    DC.EXC  ==      002H    ;ISSUE DOUBLE D INTERRUPT.

                        ;*************************************************************
```

```
                        ;*****************************************************
                        ; ASSEMBLER DIRECTIVES                               *
                        ;*****************************************************

                                .I8080          ;USE 8080 INSTRUCTION SUBSET.
                                .PABS           ;ASSEMBLE ABSOLUTE ADDRESS.
                                .PHEX           ;GENERATE INTEL HEX FORMAT.
                                .XLINK          ;SUPRESS LINKAGE OUTPUT.
0100                            .LOC    0100H   ;MODULE ADDRESS (ALTERABLE).

                        ;*****************************************************
                        ; SET STACK AND CONTROLLER ADDRESS                   *
                        ;*****************************************************

0100    31 0080     BEGIN:  LXI     SP,BSTACK       ;SET STACK POINTER.
0103    DB43                IN      D.PORT          ;INPUT STATUS PORT.
0105    E60C                ANI     DS.ASW          ;MASK FOR ADDR SWS.
0107    07                  RLC                     ;POSITION BITS.
0108    F6E4                ORI     D.BASE>8        ;OR IN BASE ADDR.
010A    67                  MOV     H,A             ;HIGH BYTE VALUE.
010B    2E00                MVI     L,0             ;LOW BYTE VALUE.
010D    22 0040             SHLD    D.ADDR          ;STORE THE ADDRESS

                        ;*****************************************************
                        ; INJECT BOOT MODULE INTO CONTROLLER                 *
                        ;*****************************************************

0110    3E01        INJECT: MVI     A,DC.MB0        ;REQUEST DD MEM BANK 0.
0112    D343                OUT     D.PORT          ;ISSUE COMMAND.
0114    01 00C0             LXI     B,IM.SZE        ;INJECTION SIZE.
0117    EB                  XCHG                    ;D.ADDR HL TO DE.

0118    21 0200             LXI     H,IM.ADR        ;INJECTION MODULE.
011B    CD 0159             CALL    BLOCK           ;BLOCK MOVE.

                        ;*****************************************************
                        ; RESET AND START THE DISK PROCESSOR                 *
                        ;*****************************************************

011E    3E80                MVI     A,DC.BGN        ;BEGIN DD PROCESSOR.
0120    D343                OUT     D.PORT          ;ISSUE  COMMAND.
0122    E3                  XTHL                    ;ALLOW DOUBLE D TIME
0123    E3                  XTHL                    ;TO START UP.

                        ;*****************************************************
                        ; WAIT FOR TASK COMPLETION                           *
                        ;*****************************************************

0124    DB43        WAIT:   IN      D.PORT          ;INPUT DD STATUS.
0126    E602                ANI     DS.HLT          ;TEST HALT* STATUS.
0128    C2 0124             JNZ     WAIT            ;WAIT TILL HALTED.
```

```
                        ;*************************************************************
                        ; SWITCH CONTROLLER MEMORY INTO SYSTEM BUS                   *
                        ;*************************************************************

012B    3E01                    MVI     A,DC.MRQ        ;REQUEST MEM (BANK 0).
012D    D343                    OUT     D.PORT          ;ISSUE COMMAND.

                        ;*************************************************************
                        ; CHECK FOR BOOTSTRAP MALFUNCTION                           *
                        ;*************************************************************

012F    2A 0040                 LHLD    D.ADDR          ;CONTROLLER ADDRESS.
0132    11 0377                 LXI     D,BL.DCS        ;ERROR CODE OFFSET.
0135    19                      DAD     D               ;SET HL POINTER.
0136    7E                      MOV     A,M             ;GET ERROR CODE.
0137    A7                      ANA     A               ;TEST REGISTER.
0138    C2 0166                 JNZ     BAD.LD          ;BAD LOAD.

                        ;*************************************************************
                        ; PERFORM BLOCK TRANSFER FROM DISK MEMORY                    *
                        ;*************************************************************

0138    2A 0040                 LHLD    D.ADDR          ;CONTROLLER ADDRESS.
013E    11 0378                 LXI     D,BL.ADR        ;LOAD ADDRESS PNTR.
0141    19                      DAD     D               ;SET HL POINTER.
0142    5E                      MOV     E,M             ;LOW ORDER ADDR.
0143    23                      INX     H               ;INCREMENT HL.
0144    56                      MOV     D,M             ;HIGH ORDER ADDR.
0145    23                      INX     H               ;REQUIRES BL.BSZ NEXT.
0146    4E                      MOV     C,M             ;LOW ORDER LENGTH.
0147    23                      INX     H               ;INCREMENT HL.
0148    46                      MOV     B,M             ;HIGH ORDER LENGTH.
0149    D5                      PUSH    D               ;USE AS JUMP ADDR.
014A    3E03                    MVI     A,DC.MB1        ;SWITCH TO MEM BANK 1.
014C    D343                    OUT     D.PORT          ;ISSUE COMMAND.
014E    2A 0040                 LHLD    D.ADDR          ;DOUBLE D MEM ADDRESS.
0151    CD 0159                 CALL    BLOCK           ;MOVE BIOS MODULE.

                        ;*************************************************************
                        ; TRANSFER CONTROL TO OPERATING SYSTEM                       *
                        ;*************************************************************

0154    3E01                    MVI     A,DC.MB0        ;SWITCH TO BANK.0
0156    D343                    OUT     D.PORT          ;ISSUE COMMAND.
0158    C9                      RET                     ;GOTO BIOS COLD ENTRY.
```

```
                        ;*************************************************************
                        ; BLOCK MOVE SUBROUTINE (Z80 BLOCK MOVE REGISTERS)    *
                        ;*************************************************************

0159    7E              BLOCK:  MOV     A,M             ;GET BYTE.
015A    23                      INX     H               ;INC POINTER
015B    EB                      XCHG                    ;GET DESTINATION.
015C    77                      MOV     M,A             ;PUT BYTE.
015D    23                      INX     H               ;INC POINTER
015E    EB                      XCHG                    ;GET SOURCE.
015F    0B                      DCX     B               ;ONE LESS TO DO.
0160    78                      MOV     A,B             ;GET HI COUNT.
0161    B1                      ORA     C               ;GET LO COUNT.
0162    C2 0159                 JNZ     BLOCK           ;FINISH LOADING.
0165    C9                      RET

                        ;*************************************************************
                        ; ERROR HAS BEEN DETECTED                                   *
                        ;*************************************************************

0166    21 016D         BAD.LD: LXI     H,ER.MSG        ;ERROR MESSAGE
0169    CD 0196                 CALL    MSG.OT          ;DISPLAY IT.
016C    76                      HLT                     ;HALT OR GOTO MONITOR.

016D    0D0A0A535953 ER.MSG:  .ASCIS  [CR][LF][LF]"SYSTEM BOOT LOAD ERROR"

                        ;*************************************************************
```

```
                        ;*****************************************************
                        ; CONSOLE LINKAGE DEFINITIONS                        *
                        ;*****************************************************

0000                    CNO.SP  ==      000H    ;OUTPUT STATUS PORT.
0004                    CNO.SB  ==      004H    ;OUTPUT STATUS BIT.
0000                    CNO.SI  ==      000H    ;OUTPUT STATUS INVERT.
0001                    CNO.DP  ==      001H    ;OUTPUT DATA PORT.

000A                    LF      ==      00AH    ;ASCII LINE FEED
000D                    CR      ==      00DH    ;CARRAIGE RETURN

                        ;*****************************************************
                        ; CONSOLE OUTPUT                                      *
                        ;*****************************************************

0186    F5              CNS.OT: PUSH    PSW             ;SAVE CHARACTER
0187    DB00            ..WAIT: IN      CNO.SP          ;INPUT STATUS
0189    EE00                    XRI     CNO.SI          ;ADJUST POLARITY
018B    E604                    ANI     CNO.SB          ;MASK STATUS BIT
018D    CA 0187                 JZ      ..WAIT          ;TRY AGAIN
0190    F1                      POP     PSW             ;RESTORE CHARACTER
0191    E67F                    ANI     07FH            ;7 BIT ASCII
0193    D301                    OUT     CNO.DP          ;SEND CHARACTER
0195    C9                      RET

                        ;*****************************************************
                        ; DISPLAY MESSAGE SUBROUTINE                         *
                        ;*****************************************************

0196    F5              MSG.OT: PUSH    PSW             ;SAVE CALLER FLAGS.
0197    7E              ..REPT: MOV     A,M             ;LOAD CHARACTER.
0198    CD 0186                 CALL    CNS.OT          ;CONSOLE OUTPUT.
019B    7E                      MOV     A,M             ;SAME CAHRACTER.
019C    23                      INX     H               ;INCREMENT POINTER.
019D    E680                    ANI     080H            ;TEST SIGN BIT.
019F    CA 0197                 JZ      ..REPT          ;ANOTHER CHARACTER.
01A2    F1                      POP     PSW             ;RESTORE FLAGS.
01A3    C9                      RET                     ;RETURN TO CALLER.

                        ;*****************************************************

0100                            .END    BEGIN
```

BAD.LD	0166	BEGIN	0100	BLOCK	0159	BL.ADR	0378
BL.BSZ	037A	BL.DCS	0377	BSTACK	0080	CNO.DP	0001
CNO.SB	0004	CNO.SI	0000	CNO.SP	0000	CNS.OT	0186
CR	000D	DC.BGN	0080	DC.EXC	0002	DC.MBO	0001
DC.MB1	0003	DC.MRQ	0001	DC.MRT	0000	DS.ASW	000C
DS.HLT	0002	D.ADDR	0040	D.BASE	E400	D.PORT	0043
ER.MSG	016D	FALSE	0000	IM.ADR	0200	IM.SZE	00C0
INJECT	0110	LF	000A	MA10	0000	MSG.OT	0196
REV.B	0001	REV.C	0000	TRUE	0001	WAIT	0124